
Motivation

• Relational Calculus and Relational Algebra were considered to be “the” database

languages for a long time

• Codd: A query language is “complete,” if it yields Relational Calculus

• However, Relational Calculus misses an important feature: recursion

• Example: A metro database with relation links:line, station, nextstation

What stations are reachable from station “Odeon”?

Can we go from Odeon to Tuileries?

etc.

• It can be proved: such queries cannot be expressed in Relational Calculus

• This motivated a logic-programming extension to conjunctive queries: datalog

Example: Metro Database Instance

links line station nextstation

4 St.Germain Odeon
4 Odeon St.Michel
4 St. Michel Chatelet
1 Chatelet Louvres
1 Louvres Palais Royal
1 Palais-Royal Tuileries
1 Tuileries Concorde

Datalog program for first query:

reach(X, X) ← links(L, X, Y)

reach(X, X) ← links(L, Y, X)

reach(X, Y) ← links(L, X, Z), reach(Z, Y)

answer(X) ← reach(‘Odeon‘, X)

Note: recursive definition

Intuitively, if the part right of “←” is true, the rule “fires” and the atom left of “←” is concluded.

Example: Ancestor-Descendant

ParentChild parent child

Homer Bart
Homer Lisa
Marge Bart
Marge Lisa
Abe Homer
Ape Abe

Query: Who are Bart’s ancestors?

Datalog program:

AncestorDescendant(X, Y) ← ParentChild(X, Y)

AncestorDescendant(X, Y) ← ParentChild(X, Z), AncestorDescendant(Z, Y)

answer(X) ← AncestorDescendant(X,′ Bart′)

Datalog Syntax

Definition. A datalog rule r is an expression of the form

R0(~x0)← R1(~x1), . . . , Rn(~xn) (1)

• where n ≥ 0,

R0, . . . , Rn are relations names, and

~x0, . . . , ~xn are vectors of variables and constants (from dom)

• every variable in ~x0 occurs in ~x1, . . . , ~xn (“safety”)

Remarks.

• The head of r, denoted H(r), is R0(~x0)

• The body of r, denoted B(r), is { R1(~x1), . . . , Rn(~xn) }

• The rule symbol “←” is often also written as “:-”

Definition. A datalog program is a finite set of datalog rules.

Datalog Programs

Let P be a datalog program.

• An extensional relation of P is a relation occurring only in rule bodies of P

• An intensional relation of P is a relation occurring in the head of some rule in P

• The extensional schema of P , edb(P ), consists of all extensional relations of P

• The intensional schema of P , idb(P ), consists of all intensional relations of P

• The schema of P , sch(P ), is the union of edb(P ) and idb(P ).

Datalog



Bases de Datos II 6

The Metro Example /1

Datalog program P on metro database scheme

M = {links : line, station, nextstation}:

reach(X, X) ← links(L, X, Y)

reach(X, X) ← links(L, Y, X)

reach(X, Y) ← links(L, X, Z), reach(Z, Y)

answer(X) ← reach(′Odeon′, X)

Here,

edb(P ) = {links} (=M),

idb(P ) = {reach, answer},
sch(P ) = {links, reach, answer}

Datalog Syntax (cntd)

• The set of constants occurring in a datalog program P is denoted as adom(P )

• Given a database instance I, we define the active domain of P with respect to I

as

adom(P, I) := adom(P ) ∪ adom(I),

that is, as the set of constants occurring in P and I

Definition. Let ν : var(r) ∪ dom→ dom be a valuation for a rule r of form (1).

Then the instantiation of r with ν, denoted ν(r), is the rule

R0(ν(~x0))← R1(ν(~x1)), . . . , Rn(ν(~xn))

which results from replacing each variable x with ν(x).

The Metro Example /2

• For the datalog program P above, we have that adom(P ) = { Odeon }

• We consider the database instance I:

links line station nextstation

4 St.Germain Odeon
4 Odeon St.Michel
4 St. Michel Chatelet
1 Chatelet Louvres
1 Louvres Palais-Royal
1 Palais-Royal Tuileries
1 Tuileries Concorde

Then adom(I) = {4, 1, St.Germain, Odeon, St.Michel, Chatelet, Louvres,

Palais-Royal, Tuileries, Concorde}

• Also adom(P, I) = adom(I).

The Metro Example /3

• The rule

reach(St.Germain, Odeon) ← links(Louvres, St.Germain, Concorde),

reach(Concorde, Odeon)

is an instance of the rule

reach(X, Y) ← links(L, X, Z), reach(Z, Y)

of P :

take ν(X) = St.Germain, ν(L) = Louvres, ν(Y ) = Odeon, ν(Z) = Concorde

Datalog Semantics

• There exist several approaches to defining the semantics:

Operational (fixpoint) approach:

Obtain query result by applying an inference procedure,

until a fixpoint is reached

Model-theoretic approach:

View rules as logical sentences, which state the query result

Proof-theoretic approach:

Obtain proofs of facts in the query result, following a proof calculus

(based on resolution)

Fixpoint Semantics

Basic idea:

“If all facts in I hold, which other facts must hold after firing the rules in P ?”

Approach:

• Define an immediate consequence operator TP (K) on db instances K.

• Start with K = I.

• Apply TP to obtain a new instance: Knew := TP (K) = I ∪ new facts.

• Iterate until nothing new can be produced.

• The result yields the semantics.

Immediate Consequence Operator

Let P be a datalog program and K be a database instance of sch(P ).

A fact R(~t) is an immediate consequence for K and P , if either

• R ∈ edb(P ) and R(~t) ∈ K, or

• there exists a ground instance r of a rule in P such that

H(r) = R(~t) and B(r) ⊆ K.

Definition. The immediate consequence operator of a datalog program P is the

mapping

TP : inst(sch(P ))→ inst(sch(P ))

where

TP (K) = { A | A is an immediate consequence for K and P }.

Example

Consider

P = { reachable(a)

reachable(Y)← arc(X, Y), reachable(X) }

where edb(P ) = {arc} and idb(P ) = {reachable}.

I = K1 = {arc(a, b), arc(b, c)}
K2 = {arc(a, b), arc(b, c), reachable(a)}
K3 = {arc(a, b), arc(b, c), reachable(a), reachable(b) }
K4 = {arc(a, b), arc(b, c), reachable(a), reachable(b), reachable(c)}

Datalog



Bases de Datos II 14

Example (cntd)

Then,

TP (K1) = {arc(a, b), arc(b, c), reachable(a)} = K2

TP (K2) = {arc(a, b), arc(b, c), reachable(a), reachable(b)} = K3

TP (K3) = {arc(a, b), arc(b, c), reachable(a), reachable(b), reachable(c) } = K4

TP (K4) = {arc(a, b), arc(b, c), reachable(a), reachable(b), reachable(c)} = K4

Thus, K4 is a fixpoint of TP .

Definition. K is a fixpoint of operator TP if TP (K) = K.

Properties

Proposition. For every datalog program P we have:

1. The operator TP is monotonic, that is, K ⊆ K′ implies TP (K) ⊆ TP (K′);

2. For any K ∈ inst(sch(P )) we have:

K is a model of ΣP if and only if TP (K) ⊆ K;

3. If TP (K) = K (i.e., K is a fixpoint), then K is a model of ΣP .

Note: The converse of 3. does not hold in general.

Fixpoint Iteration

For a datalog program P and database instance I, define the sequence (Ii)i≥0 by

I0 = I

Ii = TP (Ii−1) for i > 0.

• By monotoncity of TP , we have I0 ⊆ I1 ⊆ I2 ⊆ · · · ⊆ Ii ⊆ Ii+1 ⊆ · · ·

• For every i ≥ 0, we have Ii ⊆ B(P, I), with B(P, I) the instance of

sch(P )

• Hence, for some integer n ≤ |B(P, I)|, we have In+1 = In (=: Tω
P (I))

• It holds that Tω
P (I) = lfp(P, I) = P (I).

This can be readily implemented by an algorithm.

Example

P = { reachable(a)

reachable(Y)← arc(X, Y), reachable(X) }

I = {arc(a, b), arc(b, c)}

Then,

I0 = {arc(a, b), arc(b, c)}

I1 = T1
P (I) = {arc(a, b), arc(b, c), reachable(a)}

I2 = T2
P (I) = {arc(a, b), arc(b, c), reachable(a), reachable(b)}

I3 = T3
P (I) = {arc(a, b), arc(b, c), reachable(a), reachable(b), reachable(c)}

I4 = T4
P (I) = {arc(a, b), arc(b, c), reachable(a), reachable(b), reachable(c)}

= T3
P (I)

Thus, Tω
P (I) = lfp(P, I) = I4.

Datalog: Model-Theoretic Semantics

General Idea:

• We view a program as a set of first-order sentences

• Given an instance I of edb(P ), the result of P is a database instance of

sch(P ) that extends I and satisfies the sentences (or, is a model of the

sentences)

• There can be many models

• The intended answer is specified by particular models

• These particular models are selected by “external” conditions

Datalog

Logical Theory ΣP

• To every datalog rule r of the form R0(~x0)← R1(~x1), . . . , Rn(~xn), with

variables x1, . . . , xm, we associate the logical sentence σ(r):

∀x1, · · · ∀xm (R1(~x1) ∧ · · · ∧Rn(~xn)→ R0(~x0))

• To a program P , we associate the set of sentences ΣP = {σ(r) | r ∈ P}.

Definition. Let P be a datalog program and I an instance of edb(P ). Then,

• A model of P is an instance of sch(P ) that satisfies ΣP

• The semantics of P on input I, denoted P (I), is the least model of P

containing I, if it exists.

Example

For program P and instance I of the Metro Example, the least model is:

links line station nextstation
4 St.Germain Odeon
4 Odeon St.Michel
4 St. Michel Chatelet
1 Chatelet Louvres
1 Louvres Palais-Royal
1 Palais-Royal Tuileries
1 Tuileries Concorde

reach
St.Germain St.Germain
Odeon Odeon

· · ·
Concorde Concorde
St.Germain Odeon
St.Germain St.Michel
St.Germain Chatelet
St.Germain Louvres

· · ·
answer

Odeon
St.Michel
Chatelet
Louvres
Palais-Royal
Tuileries
Concorde

Questions

• Is the semantics P (I) well-defined for every input instance I?

• How can one compute P (I)?

Observation: For any I, there is a model of P containing I

• Let B(P, I) be the instance of sch(P ) such that

B(P, I)(R) =

 I(R) for each R ∈ edb(P )

adom(P, I)arity(R) for each R ∈ idb(P )

• Then: B(P, I) is a model of P containing I

⇒ P (I) is a subset of B(P, I) (if it exists)

• Naive algorithm: explore all subsets of B(P, I)

Elementary Properties of P (I)

Let P be a datalog program, I an instance of edb(P ), andM(I) the set of all

models of P containing I.

Theorem. The intersection
⋂

M∈M(I)M is a model of P .

Corollary.

1. P (I) =
⋂

M∈M(I)M

2. adom(P (I)) ⊆ adom(P, I), that is, no new values appear

3. P (I)(R) = I(R), for each R ∈ edb(P ).

Consequences:

• P (I) is well-defined for every I

• If P and I are finite, the P (I) is finite

Why Choose the Least Model?

There are two reasons to choose the least model containing I:

1. The Closed World Assumption:

• If a fact R(~c) is not true in all models of a database I, then infer that R(~c) is

false

• This amounts to considering I as complete

• . . . which is customary in database practice

2. The relationship to Logic Programming:

• Datalog should desirably match Logic Programming (seamless integration)

• Logic Programming builds on the minimal model semantics

Relating Datalog to Logic Programming

• A logic program makes no distinction between edb and idb

• A datalog program P and an instance I of edb(P ) can be mapped to the logic

program

P(P, I) = P ∪ I

(where I is viewed as a set of atoms in the Logic Programming perspective)

• Correspondingly, we define the logical theory

ΣP,I = ΣP ∪ I

• The semantics of the logic program P = P(P, I) is defined in terms of

Herbrand interpretations of the language induced by P :

– The domain of discourse is formed by the constants occurring in P
– Each constant occurring in P is interpreted by itself

Datalog

Herbrand Interpretations of Logic Programs

Given a rule r, we denote by Const(r) the set of all constants in r

Definition. For a (function-free) logic program P , we define

• the Herbrand universe of P , by

HU(P) =
⋃
r∈P

Const(r)

• the Herbrand base of P , by

HB(P) = {R(c1, . . . , cn) | R is a relation in P,

c1, . . . , cn ∈ HU(P), and arity(R) = n}

Example

P = { arc(a, b).

arc(b, c).

reachable(a).

reachable(Y)← arc(X, Y), reachable(X). }

HU(P) = {a, b, c}
HB(P) = {arc(a, a), arc(a, b), arc(a, c),

arc(b, a), arc(b, b), arc(b, c),

arc(c, a), arc(c, b), arc(c, c),

reachable(a), reachable(b), reachable(c)}

Grounding

• A rule r′ is a ground instance of a rule r with respect to HU(P), if r′ = ν(r)

for a valuation ν such that ν(x) ∈ HU(P) for each x ∈ var(r).

• The grounding of a rule r with respect to HU(P), denoted GroundP(r), is the

set of all ground instances of r wrt HU(P)

• The grounding of a logic program P is

Ground(P) =
⋃
r∈P

GroundP(r)

Example

Ground(P) = {arc(a, b). arc(b, c). reachable(a).

reachable(a)← arc(a, a), reachable(a).

reachable(b)← arc(a, b), reachable(a).

reachable(c)← arc(a, c), reachable(a).

reachable(a)← arc(b, a), reachable(b).

reachable(b)← arc(b, b), reachable(b).

reachable(c)← arc(b, c), reachable(b).

reachable(a)← arc(c, a), reachable(c).

reachable(b)← arc(c, b), reachable(c).

reachable(c)← arc(c, c), reachable(c). }

Herbrand Models

• A Herbrand-interpretation I of P is any subset I ⊆ HB(P)

• A Herbrand-model of P is a Herbrand-interpretation that satisfies all sentences

in ΣP,I

Equivalently, M ⊆ HB(P) is a Herbrand model if

• for all r ∈ Ground(P) such that B(r) ⊆M we have that H(r) ⊆M

Example

The Herbrand models of program P above are exactly the following:

• M1 = { arc(a, b), arc(b, c),

reachable(a), reachable(b), reachable(c) }

• M2 = HB(P)

• every interpretation M such that M1 ⊆M ⊆M2

and no others.

Logic Programming Semantics

• Proposition. HB(P) is always a model of P

• Theorem. For every logic program there exists a least Herbrand model (wrt “⊆”).

For a program P , this model is denoted MM(P) (for “minimal model”).

The model MM(P) is the semantics of P .

• Theorem (Datalog↔ Logic Programming). Let P be a datalog program and

I be an instance of edb(P ). Then,

P (I) = MM(P(P, I))

Consequences

Results and techniques for Logic Programming can be exploited for datalog.

For example,

• proof procedures for Logic Programming (e.g., SLD resolution) can be applied to

datalog (with some caveats, regarding for instance termination)

• datalog can be reduced by “grounding” to propositional logic programs

Datalog

Datalog Semantics via Least Fixpoint

The semantics of P on database instance I of edb(P ) is a special fixpoint:

Theorem. Let P be a datalog program and I be a database instance. Then

1. TP has a least (wrt “⊆”) fixpoint containing I, denoted lfp(P, I).

2. Moreover, lfp(P, I) = MM(P(P, I)) = P (I).

Advantage: Constructive definition of P (I) by fixpoint iteration

Proof-Theoretic Approach

Basic idea: The answer of a datalog program P on I is given by the set of facts

which can be proved from P and I.

Definition. A proof tree for a fact A from I and P is a labeled finite tree T such that

• each vertex of T is labeled by a fact

• the root of T is labeled by A

• each leaf of T is labeled by a fact in I

• if a non-leaf of T is labeled with A1 and its children are labeled with

A2, . . . , An, then there exists a ground instance r of a rule in P such that

H(r) = A1 and B(r) = {A2, . . . , An}

Datalog

Example (Same Generation)

P = { r1 : sgc(X, X) ← person(X)

r2 : sgc(X, Y) ← par(X, X1), sgc(X1, Y1), par(Y, Y1) }

where edb(P ) = {person, par} and idb(P ) = {sgc}

Consider I as follows:

I(person) = { 〈ann〉, 〈bertrand〉, 〈charles〉, 〈dorothy〉,

〈evelyn〉, 〈fred〉, 〈george〉, 〈hilary〉}

I(par) = { 〈dorothy, george〉, 〈evelyn, george〉, 〈bertrand, dorothy〉,

〈ann, dorothy〉, 〈hilary, ann〉, 〈charles, evelyn〉}.

Example (Same Generation)/2

Proof tree for A = sgc(ann, charles) from I and P :

r2 : par(ann, dorothy) par(charles, evelyn)

sgc(ann, charles)

r2 : par(dorothy, george)

r1 : person(george)

sgc(george, george)
par(evelyn, george)

sgc(dorothy, evelyn)

Proof Tree Construction

Different ways to construct a proof tree for A from P and I exist

• Bottom Up construction: From leaves to root

Intimately related to fixpoint approach

– Define S `P B to prove fact B from facts S if B ∈ S or by a rule in P

– Give S = I for granted

• Top Down construction: From root to leaves

In Logic Programming view, consider program P(P, I).

– This amounts to a set of logical sentences HP(P,I) of the form

∀x1 · · · ∀xm(R1(~x1) ∨ ¬R2(~x2) ∨ ¬R3(~x3) ∨ · · · ∨ ¬Rn(~xn))

– Prove A = R(~t) via resolution refutation, that is, that HP(P,I) ∪ {¬A} is

unsatisfiable.

Datalog and SLD Resolution

• Logic Programming uses SLD resolution

• SLD: Selection Rule Driven Linear Resolution for Definite Clauses

• For datalog programs P on I, resp. P(P, I), things are simpler than for general

logic programs (no function symbols, unification is easy)

• Also non-ground atoms can be handled (e.g., sgc(ann, X))

Let SLD(P) be the set of ground atoms provable with SLD Resolution from P .

Theorem. For any datalog program P and database instance I,

SLD(P(P, I)) = P (I) = T∞P(P,I) = lfp(TP(P,I)) = MM(P(P, I))

SLD Resolution – Termination

• Notice: Selection rule for next rule / atom to be considered for resolution might

affect termination

• Prolog’s strategy (leftmost atom / first rule) is problematic

Example:

child of(karl, franz).

child of(franz, frieda).

child of(frieda, pia).

descendent of(X, Y)← child of(X, Y).

descendent of(X, Y)← child of(X, Z), descendent of(Z, Y).

← descendent of(karl, X).

SLD Resolution – Termination /2

child of(karl, franz).

child of(franz, frieda).

child of(frieda, pia).

descendent of(X, Y)← child of(X, Y).

descendent of(X, Y)← descendent of(X, Z), child of(Z, Y).

← descendent of(karl, X).

SLD Resolution – Termination /3

child of(karl, franz).

child of(franz, frieda).

child of(frieda, pia).

descendent of(X, Y)← child of(X, Y).

descendent of(X, Y)← descendent of(X, Z),

descendent of(Z, Y).

← descendent of(karl, X).

SQL3

• SQL2 does not support recursive queries:

– Need to write PL/SQL or embedded SQL

• SQL3 supports recursive queries:

– WITH statement to define recursive relation

– The recursive relation is then used in a traditional SELECT FROM WHERE

query

Example: Bart’s ancestors in SQL3

Datalog Query:

AncestorDescendant(X, Y) ← ParentChild(X, Y)

AncestorDescendant(X, Y) ← ParentChild(X, Z), AncestorDescendant(Z, Y)

answer(X) ← AncestorDescendant(X,′ Bart′)

SQL3 Query:

WITH

RECURSIVE AncestorDescendent(ancestor, descendent) AS

(SELECT * FROM ParentChild)

UNION

(SELECT pc.parent, ad.descendent

FROM ParentChild pc, AncestorDescendent ad

WHERE pc.child = ad.ancestor)

SELECT ancestor

FROM AncestorDescendent

WHERE descendent = ’Bart’;

Datalog


